
Proposed change affects: UICC apps!  ME  Radio Access Network X Core Network X 
 

 
Title: ! Correction of IDL syntax errors in GenericIRPManagement{ConstDefs|System}.idl 
  
Source: ! SA5 (Nortel – Suzèle Lariven – lariven@nortel.com) 
  
Work item code: ! OAM-NIM  Date: ! 13/05/2005 
     
Category: ! F  Release: ! Rel-6 
 Use one of the following categories: 

F  (correction) 
A  (corresponds to a correction in an earlier release) 
B  (addition of feature),  
C  (functional modification of feature) 
D  (editorial modification) 

Detailed explanations of the above categories can 
be found in 3GPP TR 21.900. 

Use one of the following releases: 
Ph2 (GSM Phase 2) 
R96 (Release 1996) 
R97 (Release 1997) 
R98 (Release 1998) 
R99 (Release 1999) 
Rel-4 (Release 4) 
Rel-5 (Release 5) 
Rel-6 (Release 6) 

     Rel-7 (Release 7) 
 

  
Reason for change: ! IDL syntax error corrections in GenericIRPManagement{ConstDefs|System}.idl 
  
Summary of change: ! • Addition of missing comment separators in #endif file guards 

• Alignments with TS 32.150 Style Guide for CORBA SS IDL 
• Editorial corrections 

  
Consequences if  ! 
not approved: 

IDLs GenericIRPManagement{ConstDefs|System}.idl would fail compilation. 

  
Clauses affected: ! A.1, A.2 
  
 Y N   
Other specs !  X  Other core specifications !  
affected:  X  Test specifications  
  X  O&M Specifications  
  
Other comments: !  

Change in Annex Clause A.1 

A.1 IDL specification (file name 
“GenericIRPManagementConstDefs.idl”) 

 
//File: GenericIRPManagementConstDefs.idl 
 
#ifndef _GENERIC_IRP_MANAGEMENT_CONST_DEFS_IDL_ 
#define _GENERIC_IRP_MANAGEMENT_CONST_DEFS_IDL_ 
 
#include "<TimeBase.idl"> 
 
// This statement must appear after all include statements 
#pragma prefix "3gppsa5.org" 
 
/* ## Module: GenericIRPManagementConstDefs 
This module contains definitions commonly used among all IRPs. 
============================================================== 
*/ 
module GenericIRPManagementConstDefs 
{ 
   /* 
   Definition imported from CosTime. 
   The time refers to time in Greenwich Time Zone. 
   It also consists of a time displacement factor in the form of minutes of 
   displacement from the Greenwich Meridian. 
   */ 
   typedef TimeBase::UtcT IRPTime; 
 
   enum Signal {OK, FAILURE, PARTIALFAILURE}; 
 
   /* 
   The VersionNumber is a string that identifies the IRP specification name 
   and its version number.  See definition "IRP document version number 
   string" or "IRPVersion". 
 
   The VersionNumberSet is a sequence of such VersionNumber.  It is returned 
   by get_XXX_IRP_versions().  The sequence order has no significance. 
   */ 
   typedef string VersionNumber; 
   typedef sequence <VersionNumber> VersionNumberSet; 
 
   typedef string MethodName; 
   typedef string ParameterName; 
   typedef sequence <ParameterName> ParameterList; 
 
   /* 
   The Method defines the structure to be returned as part of 
   get_supported_operations_profile().  The name shall be the actual method 
   name (ex. "attach_push", "change_subscription_filter", etc.) 
   The parameter_list contains a list of strings.  Each string shall be 
   the actual parameter name (ex. "manager_reference", "filter", etc.) 
   */ 
   struct Method 
   { 
      MethodName name; 
      ParameterList parameter_list; 
   }; 
 
   /* 
   List of all methods and their associated parameters. 
   */ 
   typedef sequence <Method> MethodList; 
 
   typedef string NotificationName; 
   struct Notification 
   { 
       NotificationName name; 
       ParameterList parameter_list; 
   }; 
   typedef sequence <Notification> NotificationList; 
 
   /* 

   StringTypeOpt is a type carrying an optional parameter. 
   If the boolean is TRUE, then the value is present. 
   Otherwise the value is absent. 
   */ 
   union StringTypeOpt switch (boolean) 
   { 
      case TRUE: string value; 
   }; 
 
   /* 
   ShortTypeOpt is a type carrying an optional parameter. 
   If the boolean is TRUE, then the value is present. 
   Otherwise the value is absent. 
   */ 
   union ShortTypeOpt switch (boolean) 
   { 
      case TRUE: short value; 
   }; 
 
   /* 
   UnsignedShortTypeOpt is a type carrying an optional parameter. 
   If the boolean is TRUE, then the value is present. 
   Otherwise the value is absent. 
   */ 
   union UnsignedShortTypeOpt switch (boolean) 
   { 
      case TRUE: unsigned short value; 
   }; 
 
   /* 
   LongTypeOpt is a type carrying an optional parameter. 
   If the boolean is TRUE, then the value is present. 
   Otherwise the value is absent. 
   */ 
   union LongTypeOpt switch (boolean) 
   { 
      case TRUE: long value; 
   }; 
 
   /* 
   UnsignedLongTypeOpt is a type carrying an optional parameter. 
   If the boolean is TRUE, then the value is present. 
   Otherwise the value is absent. 
   */ 
   union UnsignedLongTypeOpt switch (boolean) 
   { 
      case TRUE: unsigned long value; 
   }; 
 
}; 
 
#endif // _GENERIC_IRP_MANAGEMENT_CONST_DEFS_IDL_ 

End of Change in Annex Clause A.1 
 

Change in Annex Clause A.2 

A.2 IDL specification (file name 
“GenericIRPManagementSystem.idl”) 

 
//File: GenericIRPManagementSystem.idl 
 
#ifndef _GENERIC_IRP_MANAGEMENT_SYSTEM_IDL_ 
#define _GENERIC_IRP_MANAGEMENT_SYSTEM_IDL_ 
 
#include "<GenericIRPManagementConstDefs.idl"> 
 
// This statement must appear after all include statements 
#pragma prefix "3gppsa5.org" 

module GenericIRPManagementSystem 
{ 
    exception GetInterfaceIRPNotificationProfile { string reason; }; 
    exception GetInterfaceIRPOperationProfile { string reason; }; 
    exception GetIRPVersions { string reason; }; 
 
    /* 
    Exception thrown when an unsupported optional parameter 
    is passed with information. 
    The parameter shall be the actual unsupported parameter name. 
    */ 
    exception ParameterNotSupported {  
        GenericIRPManagementConstDefs::ParameterName parameter; }; 
 
    /* 
    Exception thrown when an invalid parameter value is passed. 
    The parameter shall be the actual parameter name. 
    */ 
    exception InvalidParameter {  
        GenericIRPManagementConstDefs::ParameterName parameter; }; 
 
    /* 
    Exception thrown when a valid but unsupported parameter value is passed. 
    The parameter shall be the actual parameter name. 
    */ 
    exception ValueNotSupported {  
        GenericIRPManagementConstDefs::ParameterName parameter; }; 
 
    /* 
    Exception thrown when an unsupported optional method is called. 
    */ 
    exception OperationNotSupported {}; 
 
 
    interface  GenericIRPManagement 
    { 
        /* 
        Return the list of all supported Interface IRP versions 
        Each IRPVersion is defined by the rule in the definition 
        "IRP document version number string" or "IRPVersion" 
        (see subclause 3.1). 
        */ 
        GenericIRPManagementConstDefs::VersionNumberSet get_IRP_versions 
        ( 
        ) 
        raises (GetIRPVersions); 
 
 
      /* 
      Return the list of all supported methods and their supported 
      parameters for this Interface IRPVersion. 
      */ 
      GenericIRPManagementConstDefs::MethodList 
         get_interface_IRP_operation_profile 
      ( 
         in GenericIRPManagementConstDefs::VersionNumber this_IRP_version 
      ) 
      raises (GetInterfaceIRPOperationProfile, 
              OperationNotSupported, 
              InvalidParameter, 
              ValueNotSupported); 
 
      /* 
      Return the list of all supported notifications and their supported 
      parameters for this Interface IRPVersion. 
      */ 
      typedef  GenericIRPManagementConstDefs::NotificationList NotificationList; 
 
      NotificationList get_interface_IRP_notification_profile 
      ( 
         in GenericIRPManagementConstDefs::VersionNumber this_IRP_version 
      ) 
      raises (GetInterfaceIRPNotificationProfile, 
              OperationNotSupported, 
              InvalidParameter, 
              ValueNotSupported); 
    }; 

}; 
 
#endif // _GENERIC_IRP_MANAGEMENT_SYSTEM_IDL_ 

End of Change in Annex Clause A.2 
End of Document 

Annex B (informative): 
Change history 

Change history 
Date TSG # TSG Doc. CR Rev Subject/Comment Old New 
Dec 2004 S_26 SP-040795 -- -- Submitted to SA#26 for Approval 1.0.0 6.0.0 

